

ACTIVI... TO 7 : ZENLIGNE

Une aide à l'apprentissage de l'addition dans \mathbb{Z} en cinquième

I.R.E.M. de Grenoble

Bernard CAPPONI
Collège Le Vergeron, Moirans
Philippe CLAROU
Collège Camille Vernet, Valence


ZENLIGNE est un programme pour les enfants de cinquième. Il leur permet de s'entraîner au calcul de sommes d'entiers relatifs présentées en ligne sans parenthèses, pendant le temps nécessaire à chacun pour que des calculs du type $-3 + 4 - 8$ soient bien acquis avant l'entrée en quatrième.

UTILISATION.

Ce programme propose des séries d'exercices : l'utilisateur peut choisir le nombre d'exercices de chaque série et le nombre de termes dans chaque exercice.

A la mise en route il est proposé de choisir le nombre d'exercices par série et le nombre de termes dans l'exercice. Pour faire ce choix il faut presser sur **C** et les choix sont proposés. Sinon en appuyant sur **ENTREE** la machine choisit par défaut **10** exercices avec **3** termes chacun.

L'écran se présente alors ainsi :



1 est le numéro de l'exercice en cours, 10 le nombre d'exercices à réaliser.

L'élève doit taper sa réponse puis **ENTREE** (la machine accepte seulement les chiffres et les signes + et -).

La flèche ← permet de corriger avant d'avoir tapé **ENTREE**).

La réponse de l'élève est analysée, et suivant les cas les messages suivants apparaissent sur l'écran.

EXACT

ou

INEXACT le résultat exact est -4.

Après avoir observé la réponse s'il le veut, l'élève tape **ENTREE** pour poursuivre et un nouvel exercice lui est proposé. Et ainsi de suite jusqu'à la fin de la série où un bilan est alors affiché.

On obtient une autre série d'exercices en pressant **ENTREE**. On peut aussi choisir d'autres paramètres (longueur de la série et nombre de termes) en pressant sur **M**.

On peut également choisir de sortir du programme en tapant **F**, on obtient alors un bilan de l'ensemble des exercices avec le pourcentage de réussites.

Chacun a ainsi la possibilité d'une pratique aussi longue que nécessaire pour parvenir à une bonne acquisition de ce savoir faire.

COMMENT L'AVONS NOUS UTILISE ?

Nous avons proposé **ZENLIGNE** à des élèves de cinquième au moment où nous avons abordé les simplifications d'écritures dans des additions et soustractions d'entiers relatifs sans utiliser de parenthèses. Nous disposions de cinq TO 7 dans une salle spécialisée communiquant avec une salle de cours ordinaire. Cette organisation nous a permis de faire travailler les élèves avec ce logiciel pendant que le reste de la classe travaillait à d'autres activités sur fiches.

Nous avons également utilisé **ZENLIGNE** pour le soutien d'élèves ayant des difficultés particulières et pour lesquels était nécessaire davantage d'entraînement. Il a parfois été possible, à la demande des élèves, de les faire travailler pendant des heures de permanence avec l'aide du documentaliste.

Nous avons dans tous les cas, utilisé ce logiciel comme une aide à l'entraînement dans une situation qui avait déjà été abordée en classe et non pas dans une situation d'apprentissage.

Nous avons presque toujours placé deux enfants ensemble devant la même console. L'ordinateur propose alors une tâche à réaliser : les deux enfants réfléchissent, calculent, confrontent leurs résultats et proposent une réponse que la machine valide. En cas d'échec, le résultat exact est affiché en même temps que la question et que le résultat qui a été donné : ceci ouvre la possibilité d'une réflexion sur l'exercice effectué avant d'en recommencer un autre.

QUELQUES REMARQUES SUR LE PROGRAMME.

Nous avons essayé d'écrire un programme clair et commenté afin de le rendre lisible par des enseignants ayant reçu une formation élémentaire en informatique.

Nous donnons simplement ici un commentaire plus détaillé à propos de la procédure de saisie (ligne 1 500 à 1 610 du listing).

Cette procédure est utilisée pour :


- entrer les paramètres du programme (nombre d'exercices et de termes dans l'exercice) avec des caractères de taille normale ;
 - entrer la réponse de l'élève avec des caractères double hauteur et double largeur. Nous avons utilisé l'instruction **INKEY\$** dans cette procédure pour répondre à des impératifs que nous nous étions fixés. En effet :
 - on peut saisir et afficher des caractères double taille alors que l'instruction **INPUT** entraîne automatiquement l'affichage en taille normale ;
 - on peut effacer sa réponse avant de la valider en tapant la touche ← (recul curseur et effacement du dernier caractère tapé) ;
 - on contrôle les caractères tapés au clavier en refusant tout ce qui n'est pas un chiffre ou l'un des deux signes + ou -. Ceci évite un grand nombre de fautes de frappe ;
 - on contrôle la longueur maximum de la réponse donnée (avec la variable **LIM**) ;
 - on fait tourner à vide le générateur de nombres aléatoires durant toute cette procédure.
- On produit ainsi un tirage vraiment au hasard.

Avant d'appeler cette procédure on fournit la taille des caractères (variable **TAI**) et la longueur maximum de la chaîne à saisir (variable **LIM**).

Les caractères sont saisis un par un dans une variable **I\$** dont le contenu est analysé et permet de construire la chaîne **IR\$** caractère par caractère. C'est cette variable **IR\$** qui est produite par cette procédure.

Voici un organigramme de cette procédure (chaque numéro correspond à la ligne du programme **BASIC**). Nous donnons ensuite la liste des variables utilisées et la liste du programme en **BASIC TO 7**.

Procédure de saisie (ZENLIGNE)



```

10 '*****
20 'ZENLIGNE
30 'IREM DE BRENOBLE B.CAPPONI P.CLAROU
40 'GROUPE 1er Cycle   mai 84
50 'BP 41
60 '38402 Saint Martin d'Herès
70 '*****

```

PAGE TITRE

```

100 PRINTCHR$(20);:'effacement curseur pour couverture
110 CLS:SCREEN2,0,0:ATTRB1,1:LOCATE 12,11:PRINT"ZENLIGNE":ATTRB0,0
120 LOCATE 8,14:PRINT"Addition dans Z en ligne"
130 GOSUB 910
140 CLS
150 PRINT CHR$(17):'retour curseur

```

CHOIX POUR L'EXERCICE

```

160 LOCATE2,12:PRINT"POUR CHOISIR   -le nombre d'exercices"
170 PRINT"                               -le nombre de termes"
180 LOCATE 18,16 :COLOR 5:PRINT"taper ";:COLOR0,3:PRINT"C":COLOR 2,0:
190 LOCATE18,22:COLOR5:PRINT"sinon ";:COLOR 0,3:PRINT"ENTREE":COLOR2,0
200 IR$=INPUT$(1)
210 IF IR$="C"ORIR$="c"THEN CLS:ELSE          TER=3:NB=10:GOTO 400
300 'CHOIX pour l'exercice
310 'choix nombre de termes
320 LOCATE 2,12:PRINT"combien de termes (2,3 ou 4)? ";:COL=POS(0):LI6=CSRLIN
330 LIM=1:GOSUB 1500:TER=VAL(IR$)
340 IF TER=2 OR TER=3 OR TER=4 THEN 350 ELSE LOCATE 2,12:PRINTSPC(38):GOTO 320
350 'choix du nombre d'exercices
360 CLS:LOCATE 2,12:PRINT"Combien d'exercices ? ";:COL=POS(0):LI6=CSRLIN
370 LIM=2:GOSUB 1500'saisie inkey$
380 IF IR$="" THEN 370 ELSE NB=VAL(IR$)

```

DEROULEMENT DE L'EXERCICE

```

410 FL=0
420 FOR RET=1 TO NB
430 EXA=0:CLS:TAI=1:BOX(0,56)-(319,120),1:LOCATE 30,5:PRINTRET;"   "NB;:LOCATE 1,11
440 FOR I=1 TO TER
450 ATTRB1,1
460 X(I)=INT(21*RND)-10
470 IF X(I)=0 THEN 460
480 EXA=EXA+X(I):X$=STR$(X(I)) :LX=LEN(STR$(X(I))): X$=RIGHT$(X$,LX-1)
490 IF I=1 THEN IFX(I)>0 THEN PRINTX$;CHR$(9); ELSE PRINT"-";CHR$(9);X$;CHR$(9);
500 IF I>1 THEN IFX(I)>0 THEN PRINT"+";CHR$(9);X$;CHR$(9); ELSE PRINT"-";CHR$(9);X$;CHR$(9);
510 NEXT
520 PRINT"= ";:COL=POS:LI6=CSRLIN
530 LIM=4:GOSUB 1500:REP=VAL(IR$):ATTRB0,0
540 LOCATE 0,17 :COLOR5
550 IF REP=EXA THEN PRINT"Exact" ELSE PRINT"Inexact ":COLOR2: LOCATE 0,19:PRINT"Le resultat exact
   est ";:COLOR5:ATTRB1,1:PRINTEXA :FL=FL+1:ATTRB0,0
560 COLOR2: GOSUB 910
570 NEXT
580 CLS:COLOR5:LOCATE 0,15:PRINTNB;" Exercice";:IF NB>1 THEN PRINT"s"
590 PRINT:PRINTFL ;" Erreur";:IF FL>1 THEN PRINT"s"
600 GOSUB 910
610 FT=FT+FL:COMP=COMP+NB'compteurs

```

RELANCE APRES UNE SERIE D'EXERCICES

```

640 CLS:LOCATE 0,8:PRINT"Pour terminer .....taper ";:COLOR4,3:PRINT"F";:COLOR 2,0
650 LOCATE 0,11:PRINT"pour modifier .....taper ";:COLOR4,3:PRINT"M";:COLOR2,0
660 LOCATE 0,15:PRINT"pour recommencer.....taper ";:COLOR 0,3:PRINT"ENTREE";:COLOR2,0
670 I$=INPUT$(1):PRINTI$
680 IF ASC(I$)=13 THEN 400
690 IF I$="f" OR I$="F" THEN 800
700 IF I$="m" OR I$="M" THEN CLS:GOTO300 ELSE 670

```

BILAN FINAL

```

810 CLS:COLOR3:LOCATE 13,2:PRINT"BILAN FINAL"
820 LOCATE 3,8:PRINT"OMBRE EXERCICES : ";:COLOR 5:PRINTCOMP:COLOR3
830 LOCATE 3,10:PRINT"OMBRE ERREURS : ";:COLOR 5:PRINTFT:COLOR3
840 IF COMP<>0 THEN LOCATE 3,12:PRINT"POURCENTAGE REUSSITE : ";:COLOR 5:PRINTUSING"##.#";(COMP-FT)/C
OMP*100;PRINT" %":COLOR3
850 COLOR5,3:LOCATE 0,17:PRINT"Au revoir":LOCATE 0,20:PRINT"IREM DE GRENOBLE"
860 PRINT"BP 41"
870 PRINT"38402 Saint Martin d'Herès"
880 COLOR 2,0 :END

```

PROCEDURE DE RELANCE

```

910 LOCATE 12,22:PRINT"pour continuer taper ";:COLOR 0,3:PRINT"ENTREE":COLOR2,0
920 LIM=0:GOSUB 1500:RETURN

```

PROCEDURE DE SAISIE

```

1510 PRINTCHR$(20):I$=INKEY$:IR$=""
1520 I$=INKEY$:INIT=RND:IF I$="" THEN 1520
1530 IFASC(I$)=8 THEN IF IR$="" THEN 1520 ELSE IR$=LEFT$(IR$,LEN(IR$)-1):LOCATE POS-TAI-1,CSRLIN :
PRINTSPC(1);:LOCATE POS-TAI-1,CSRLIN:GOTO1520
1540 IFASC(I$)=13 THEN RETURN
1550 IF LEN(IR$)>LIM-1 THEN 1520
1560 IF ASC(I$)=45 OR ASC(I$)=43 THEN 1580
1570 IFASC(I$)<48 OR ASC(I$)>57 THEN 1520
1580 IR$=IR$+I$
1590 LOCATECOL,LIG:PRINTIR$;
1600 GOTO 1520
1610 'fin procedure de saisie

```

LISTE DES VARIABLES UTILISEES

COL	position du curseur sur la ligne
COMP	compteur du nombre total d'exercices réalisés
EXA	résultat exact de chaque exercice
FL	nombre d'erreurs dans une série d'exercices
FT	nombre d'erreurs cumulé pour le bilan final.
I\$	saisie d'un caractère dans la procédure 1500.
IR\$	chaîne sortant de la procédure de saisie (1500).
LIG	ligne du curseur.
LIM	longueur maximum de la chaîne à saisir dans la procédure de la ligne 1500.
LX	variable de travail pour l'affichage
NB	nombre d'exercices (limité à 99).
REP	réponse de l'élève à chaque exercice
RET	numéro de l'exercice en cours
TAI	taille des caractères TAI-0 : petit caractère .TAI-1 : grand caractère
TER	nombre de termes de l'exercice proposé (limité à 2,3 ou 4).
X(1)	tableau contenant les termes de l'exercice
X\$	variable de travail pour construire l'affichage de la question